
EctoTemp
    

EctoTemp provides macros and helper functions for creating and utilizing temporary tables in
Postgres. Among other use cases, this library can be utilized to test data migrations (where the
underying table may change over time) or ecto type extensions in isolation, where no one persistent
table exist.
The temporary tables created by this library are transaction local, using ON COMMIT DROP. The
PostgreSQL documentation on creating
tables describes this setting as follows:
The temporary table will be dropped at the end of the current transaction block. When used on a
partitioned table, this action drops its partitions and when used on tables with inheritance
children, it drops the dependent children.

For use in ExUnit, this means that any temporary tables created in the context of the
Ecto.Adapters.SQL.Sandbox are scoped to the test's database connection, and are deleted at the end
of each test.

  
    
  
  Usage


After adding :ecto_temp to a project's Mix dependencies, a test can use EctoTemp, repo: <My.Repo>. Temporary tables are defined with EctoTemp.Macros.deftemptable/3 and
EctoTemp.Macros.column/2.
Temp tables are actually created in the database with EctoTemp.Macros.create_temp_tables/0.
defmodule MyTest do
  use Test.DataCase, async: true
  use EctoTemp, repo: MyProject.Repo

  deftemptable :things_temp do
    column :data, :string, null: false
    column :data_with_default, :string, default: "default value"
    deftimestamps()
  end

  setup do
    create_temp_tables()
    :ok
  end
end
In tests, EctoTemp.Factory.insert/3 can be used to insert data into a temporary table. As a macro,
it must be used via import or require.
defmodule MyTest do
  # ...

  require EctoTemp.Factory

  test "inserts data into a temp table" do
    Factory.insert(:things_temp, data: "stuff")
  end
end
An example of a real project using EctoTemp to test an Ecto.Type extensions can be found in the
ecto_date_time_range
project.


  

    
Data Migrations
    

Database data migrations face at least two primary problems with regard to testing: database schemas
will change in the future; application modules will change in the future. Tests that rely on either
face a high likelihood of failing.
Ecto provides a mechanism for overriding the database table used by a query, via Ecto.put_meta/2
with the :source attribute, or via {table, Schema} when making queries.

  
    
  
  Example Data Migration


Imagine a data migration that takes a bunch of things, modifies one or more fields, and creates
other related things.
This data migration can define its own schema modules, representing the schema at the time of
migration. At any place a database interaction occurs, the table name for each entity must be
specified. To ensure that this is the case, the schemas define their tables as
replace_me_at_runtime, so that code that forgets to override it will blow up.
defmodule Core.DataMigrations.MyMigration do
  import Ecto.Query

  defmodule Thing do
    use Ecto.Schema
    import Ecto.Changeset

    schema "replace_me_at_runtime" do
      field :data, :string
      timestamps()
    end

    @doc """
    Note that when constructing a query, `{table_name, Thing}` is used
    in place of `Thing`. This overrides the database table used by the
    query.
    """
    def all(table_name),
      do: from(things in {table_name, Thing})

    @doc """
    When modifying data, `Ecto.put_meta/2` is utilized to enfore the
    table name.
    """
    def update_changeset(thing, table_name) do
      thing
      |> Ecto.put_meta(source: table_name)
      |> change(data: thing.data <> "-ish")
    end
  end

  defmodule OtherThing do
    use Ecto.Schema
    import Ecto.Changeset

    schema "replace_me_at_runtime" do
      field :thing_id, :bigint
      timestamps()
    end

    def insert_changeset(thing, table_name) do
      %OtherThing{}
      |> Ecto.put_meta(source: table_name)
      |> change(thing_id: thing.id)
    end
  end

  @doc """
  Table names are injected by the tests, but default to the real tables used by
  the application.
  """
  def run(things_table \\ "things", other_things_table \\ "other_things") do
    Core.Repo.transaction(fn ->
      Thing.all(things_table)
      |> Repo.stream()
      |> Stream.map(&Thing.update_changeset(&1, things_table))
      |> Stream.map(&update_thing/1)
      |> Stream.map(&OtherThing.insert_changset(&1, other_things_table))
      |> Stream.map(&insert_other_thing/1)
      |> Stream.run()
    end, timeout: 60_000)
  end

  def update_thing(thing_changeset),
    do: thing_changeset |> Core.Repo.update!()

  def insert_other_thing(other_thing_changeset),
    do: other_thing_changeset |> Core.Repo.insert!()
end

  
    
  
  Example Test


The test can rely on the schemas defined by the migration, and inject temporary tables at runtime:
defmodule Core.DataMigrations.MyMigrationTest do
  use Test.DataCase, async: true
  use EctoTemp, repo: Core.Repo

  alias Core.DataMigrations.MyMigration, as: Migration
  require EctoTemp.Factory

  deftemptable :temp_things do
    column :data, :string, null: false
    deftimestamps()
  end

  deftemptable :temp_other_things do
    column :thing_id, :integer, null: false
    deftimestamps()
  end

  setup do
    create_temp_tables()
    :ok
  end

  describe "run" do
    test "does stuff" do
      Factory.insert(:temp_things, data: "stuff")
      assert {"temp_other_things", Migration.OtherThing} |> Core.Repo.all() == []

      # inject tables
      Migration.run("temp_things", "temp_other_things")

      [thing] = {"temp_things", Migration.Thing} |> Core.Repo.all()
      assert thing.data == "stuff-ish"

      assert [other_thing] =
          {"temp_other_things", Migration.OtherThing} |> Core.Repo.all()

      assert other_thing.thing_id == thing.id
    end
  end
end


  

    
Change Log
    


  
    
  
  Unreleased



  
    
  
  2.0.1


	Add sponsorship link.


  
    
  
  2.0.0


	Disambiguate Factory macro variations into separate heads.
	Test against Elixir 1.18.
	Breaking change: Drop support for Elixir 1.15.


  
    
  
  1.2.0


	require Elixir 1.15 or greater
	test against the latest version of Erlang and Elixir


  
    
  
  1.1.0


	Drop support for Elixir 1.13 and older.


  
    
  
  1.0.1


	Update deps.


  
    
  
  1.0.0


	Add EctoTemp.Factory.uuid/0.
	Do not automatically append _temp to table names.


  
    
  
  0.1.3


	Add :primary_key info to documentation.


  
    
  
  0.1.2


	Add overview, data migrations to documentation.


  
    
  
  0.1.1


	:ecto and :ecto_sql are required in any environment that uses :ecto_temp.
	Use Apache 2.0 license.
	Updated project URL.


  
    
  
  0.1.0


	Initial release.



  

    
License
    

Copyright 2022 synchronal.dev
Licensed under the Apache License, Version 2.0 (the "License"); you may not use this file except in
compliance with the License. You may obtain a copy of the License at
 http://www.apache.org/licenses/LICENSE-2.0
Unless required by applicable law or agreed to in writing, software distributed under the License is
distributed on an "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or
implied. See the License for the specific language governing permissions and limitations under the
License.


  

    
EctoTemp 
    



      
EctoTemp is use'd to set up a module for managing temp tables. Once set up, macros such as
EctoTemp.Macros.deftemptable/3, EctoTemp.Macros.column/3, and macros in EctoTemp.Factory
may be used.

  
    
  
  Examples


defmodule MyTest do
  use MyProject.DataCase
  use EctoTemp, repo: MyProject.Repo

  require EctoTemp.Factory
  alias EctoTemp.Factory

  deftemptable :things_temp do
    column :data, :string, null: false
    column :data_with_default, :string, default: "default value"
    deftimestamps()
  end

  deftemptable :other_things_temp, primary_key: false do
    column :id, :uuid, null: false
    column :data, :string, null: false
    column :data_with_default, :string, default: "default value"
    deftimestamps()
  end

  setup do
    create_temp_tables()
    :ok
  end

  test "insert records" do
    Factory.insert(:things_temp, data: "stuff")
  end
end

      


      
        Summary


  
    Functions
  


    
      
        __using__(opts)

      


        Imports EctoTemp into a module.



    





      


      
        Functions

        


  
    
      
    
    
      __using__(opts)


        (macro)


        
          
        

    

  


  

Imports EctoTemp into a module.

  
    
  
  Options


	:repo - required - the module defining your Ecto.Repo callbacks.


  
    
  
  Example


defmodule MyModule do
  use EctoTemp, repo: MyProject.Repo
end

  


        

      


  

    
EctoTemp.Factory 
    




      
        Summary


  
    Functions
  


    
      
        insert(table)

      


        Inserts values into a temporary table.



    


    
      
        insert(module, table)

      


    


    
      
        insert(module, table, params)

      


    


    
      
        uuid()

      


        Generates a UUID that may be inserted directly into a :uuid field. Use this in preference of
Ecto.UUID.generate/0 or Ecto.UUID.bingenerate/0, as those will not dump their string values
to the binary format expected by postgres.



    





      


      
        Functions

        


  
    
      
    
    
      insert(table)


        (macro)


        
          
        

    

  


  

Inserts values into a temporary table.

  
    
  
  Params:


	struct (optional) - a struct defined the schema used by the data migration.
	table_name
	attrs (optional) - a keyword list of attributes to insert


  
    
  
  Notes:


	If not given a struct, and the temporary table has a primary key, then we return the id of the inserted row.
	If given a struct, and the temporary table has a primary key, then we do a Repo.get using the id of the
inserted row, and return the result as a struct.
	If the temporary table has no primary key, then we return the list of values returned by postgres. This list
is probably ordered by the order in which the columns are defined on the temp table???


  
    
  
  Examples


import EctoTemp.Factory

insert(:thing_with_no_primary_key) == []
insert(:thing_with_no_primary_key, some_thing: "hi") == ["hi"]
insert(:thing_with_primary_key) == 1
insert(:thing_with_primary_key, some_thing: "hi") == 2
%MyDataMigration.Cycle{id: 1} = insert(MyDataMigration.Cycle, :cycles, started_at: ~N[2020-02-03 00:00:00])

  



  
    
      
    
    
      insert(module, table)


        (macro)


        
          
        

    

  


  


  



  
    
      
    
    
      insert(module, table, params)


        (macro)


        
          
        

    

  


  


  



  
    
      
    
    
      uuid()



        
          
        

    

  


  

      

          @spec uuid() :: binary()


      


Generates a UUID that may be inserted directly into a :uuid field. Use this in preference of
Ecto.UUID.generate/0 or Ecto.UUID.bingenerate/0, as those will not dump their string values
to the binary format expected by postgres.

  


        

      


  

    
EctoTemp.Macros 
    




      
        Summary


  
    Functions
  


    
      
        column(name, type, opts \\ [])

      


        Add a column to a table definition.



    


    
      
        create_temp_tables()

      


        Runs through previously defined @ecto_temporary_tables to insert temporary tables.
This should be called in a setup block, which needs to be defined after all
deftemptable definitions.



    


    
      
        deftemptable(table_name, opts \\ [], list)

      


        Creates a temporary table that will be rolled back at the end of the
current test transaction. If the table name is given as :things, then
the actual temporary table will be created as things. As this may
overlap with existing tables, it is recommended to use a temp prefix
of suffix when defining temp tables.



    


    
      
        deftimestamps()

      


        Adds inserted_at and updated_at to a table definition.



    





      


      
        Functions

        


    

  
    
      
    
    
      column(name, type, opts \\ [])


        (macro)


        
          
        

    

  


  

      

          @spec column(name :: atom(), type :: atom(), opts :: keyword()) :: Macro.t()


      


Add a column to a table definition.
This must be called within a deftemptable block, or a CompileError will be raised.

  
    
  
  Opts


	name	type	default	description
	default	term()		Provides a default value when none is provided.
	null	boolean	true	Determines whether null values are allowed in a column.


  



  
    
      
    
    
      create_temp_tables()


        (macro)


        
          
        

    

  


  

      

          @spec create_temp_tables() :: Macro.t()


      


Runs through previously defined @ecto_temporary_tables to insert temporary tables.
This should be called in a setup block, which needs to be defined after all
deftemptable definitions.

  



    

  
    
      
    
    
      deftemptable(table_name, opts \\ [], list)


        (macro)


        
          
        

    

  


  

Creates a temporary table that will be rolled back at the end of the
current test transaction. If the table name is given as :things, then
the actual temporary table will be created as things. As this may
overlap with existing tables, it is recommended to use a temp prefix
of suffix when defining temp tables.

  
    
  
  Examples


deftemptable :person_temp do
  column :thing_id, :integer, null: false
  deftimestamps()
end

deftemptable :thing_temp, primary_key: false do
  column :description, :string, null: false
  column :comment, :string
end

  
    
  
  Opts


	name	type	default	description
	primary_key	boolean	true	When true, adds an :id field of type :bigserial


  



  
    
      
    
    
      deftimestamps()


        (macro)


        
          
        

    

  


  

      

          @spec deftimestamps() :: Macro.t()


      


Adds inserted_at and updated_at to a table definition.
This must be called within a deftemptable block, or a CompileError will be raised.

